Total Return Swap with Mathematica 10

Total return swap, which is probably better known under its abbreviation **TRS**, is another popular derivative contract that was developed from a traditional swap format to enable synthetic replication of financial asset. In TRS one side pays regularly a known rate (fixed or floating money-market index)and receives total return (all income +/- capital gain) of the underlying asset. The rate leg is known as 'financing', the other as 'return'. The asset can be any tradable instrument – equity, bond, commodity or financial index. TRS format therefore enables to gain exposure to financial instrument without actually owning it. Thanks to this, TRS have become popular tool for exposure access by many opportunistic players such as hedge funds and leverage-seeking investors.
Although TRS has been classified as *credit derivative product*, this categorisation is too narrow. The contract transfers all risk factors of an underlying asset (market and credit) and therefore requires broader definition as *replication product*. Equity and commodity swaps are actually a variation of TRS in their respective markets.
Valuation of TRS follows the principles applicable in the traditional swap market. The TRS is priced at inception by determining the fixed rate (or margin on the floating index) such that the value of the both legs is identical. **Mathematica 10**, similar to all previous tasks, provides right functionality and tools to price not only simple but also more complex TRS.

Illustration scheme of the TRS:



As the above transaction scheme demonstrates, TRS transforms the physical asset into synthetic one where the investor effectively gains access to the economic performance of the asset without the necessity to purchase it.

This offers number of advantages (i) balance sheet utilisation, (ii) access to the restricted markets, (iii) currency risk elimination and (iv) investment horizon adjustment (TRS maturity and the asset maturity can be different, assuming that asset maturity > TRS contract maturity).

Consider a case where investor wants to gain an access to the emerging markets bonds universe, but is limited by currency, liquidity concerns of the EM universe or physical ownership constrains. (S)he can enter into a TRS contract on the *JP Morgan USD Emerging Markets index* which is tradable and liquid instrument that tracks performance of the USD-denominated EM bonds. The investment horizon is set for 1Y and the financing leg is supposed to pay USD 1M Libor + margin. The objective is to price the TRS – i.e. determining a fair-value margin such that the NPV of both legs are identical.

1) Financing leg:

We build the Libor curve first:

```
zrates = {0.0022, 0.0039, 0.0067, 0.0086, 0.0101, 0.0121, 0.0134, 0.0145};
times = {0.25, 0.5, 1, 1.5, 2, 3, 4, 5};
zrobj = TemporalData[zrates, {times}];
```

And define the discount factor and Libor as follows:

```
intrate = Interpolation[zrobj["Path"], Method → "Spline"];
DF[x_] := Exp[-intrate[x] * x]
```

$$\texttt{libor[p\_, q\_]} := \frac{1}{q-p} * \left(\frac{\texttt{DF[p]}}{\texttt{DF[q]}} - 1\right)$$

The 1M Libor curve looks as follows:

```
ListLinePlot[Table[{i, libor[i - 1/12, i]}, {i, 1/12, 2, 1/12}],
 PlotLabel → "USD 1M Libor", PlotTheme → "Web"]
```



These rates will be paid on the financing leg over the period of the contract.

2) Total return leg:

The objective is to determine the total return on the EMB index in one year.

We first look at the past 3Y history:

```
hdata = TimeSeries[FinancialData["EMB", "January 1, 2012"]]
```

```
TimeSeries[ Time: 03 Jan 2012 to 31 Oct 2014
            Data points: 713 ]
```

The index has been quite volatile:

```
DateListPlot[hdata, Filling → Axis,
  PlotLabel → "US$ Emerging markes index"]
```



Although under the TRS the return is typically determined from the known parameters, we ca do better and estimate the future index value thought simulation. Given the nature of the underlying assets, we select the <u>CIR stochastic process</u> (mean-reverting square-root diffusion) for this task.

a) Calibrate the CIR process to the past data
b) Run the simulation for 1Y period
c) Get the expectation and total return = $\frac{P(t)}{P(0)} - 1$

```
tsw = TimeSeriesWindow[hdata, {{2014, 1, 1}, Automatic}];

estCIR = EstimatedProcess[tsw["Values"], CoxIngersollRossProcess[a, b, c, tsw[lastdate]]]

CoxIngersollRossProcess[116.149, 0.0325521, 0.00892425, 114.68]

simpaths = RandomFunction[estCIR, {0, 360, 1}, 1000];
simobj = TemporalData[simpaths["ValueList"], {lastdate}];
simindex = DateListPlot[simobj["Path", Range[15]], PlotStyle → Opacity[0.9],
  PlotLabel → "Esrimated index with the CIR model"]
```



The calibrated CIR process returns realistic estimates of the future paths. From here we can easily obtain the expected value of the index for the next 12 months:

```
meanCIR = TimeSeriesThread[Mean, simobj];
DateListPlot[meanCIR, PlotLabel → "Expected value if the index", PlotStyle → Magenta,
  Filling → Axis, FillingStyle → Directive[Opacity[0.25], Gray]]
```



Giver the recent history we expect the index to increase its value over the next 12 months.
Once the expected value is known, we can get the total return from initial and final value of the index:

```
estvalues = {First[meanCIR["Values"]], Last[meanCIR["Values"]]}

{114.68, 116.21}

totreturn = estvalues[[2]] / estvalues[[1]] - 1

0.0133419
```

We estimate the index return in the range 1.33% (ignoring intermediate flows).

Knowing the return leg performance, we can now price the TRS easily by solving the swap equation for the unknown margin:

$$TR * DF[1y] = \sum(Libor[i] + margin) * dt * DF[i]$$

```
soln = FindRoot[Sum[(libor[i - 1 / 12, i] + m) * 1 / 12 * DF[i], {i, 1 / 12, 1, 1 / 12}]
    == totreturn * DF[1], {m, 0.001}]

{m → 0.00659301}
```

The fair-value margin on the financing leg is 0.66%.

The financing leg flows:

```
BarChart[Table[(libor[i - 1 / 12, i] + m /. soln) * 1 / 12 * DF[i], {i, 1 / 12, 1, 1 / 12}],
 ChartStyle → "Rainbow", ChartLabels → Table[i, {i, 1, 12}],
 PlotLabel → "Financing leg discounted flows"]
```



The cash flow table:

```
TableForm[Table[{libor[i - 1 / 12, i], m /. soln, DF[i]}, {i, 1 / 12, 1, 1 / 12}],
 TableHeadings → {Table[busday[startdate, i], {i, 1, 12}], {"1M Libor", "Margin", "DF"}}]
```

	1M Libor	Margin	DF
Wed 10 Dec 2014	0.00098691	0.00659301	0.999918
Mon 12 Jan 2015	0.00221389	0.00659301	0.999733
Tue 10 Feb 2015	0.00339993	0.00659301	0.99945
Tue 10 Mar 2015	0.00453819	0.00659301	0.999072
Fri 10 Apr 2015	0.00562183	0.00659301	0.998604
Mon 11 May 2015	0.00664399	0.00659301	0.998052
Wed 10 Jun 2015	0.00759781	0.00659301	0.99742
Fri 10 Jul 2015	0.00847646	0.00659301	0.996716
Mon 10 Aug 2015	0.00927306	0.00659301	0.995947
Thu 10 Sep 2015	0.00998077	0.00659301	0.995119
Tue 13 Oct 2015	0.0105927	0.00659301	0.994241
Tue 10 Nov 2015	0.0111021	0.00659301	0.993322

<u>In short</u>: the valuation of TRS in Mathematica 10 is quick and simple and the platform naturally extends its reach beyond the horizon offered by standard valuation techniques.